Fitting Wave Height Data to a Probability Distribution

Introduction

The University of Maine records real-time accelerometer data from buoys deployed in the Gulf of Maine and the Caribbean (http://gyre.umeoce.maine.edu/buoyhome.php). The data can be downloaded from their website, and include the significant wave height recorded at regular intervals for the last few months.

This application

- downloads accelerometer data for Buoy PR206 (located just off the coast of Puerto Rico at a latitude of 18° 28.46' N and a longitude of 66° 5.94' W),
- fits the significant wave height to a Weibull distribution via two methods: maximum likelihood estimation and moment matching,
- and plots the fitted distributions on top of a histogram of the experimental data.

The Google Maps component below gives the location of buoy PR206.

Latitude 18.47433

Longitude -66.099

Zoom Level

Map Type satellite

Download and Plot Significant Wave Height Data in a Histogram

```
> restart :
> with( plots ) : with( Statistics ) : with( Optimization ) :
```

```
> url
      := "http://gyre.umeoce.maine.edu/data/gomoos/buoy/php/view_csv_file.php?ncfile=
    /data/gomoos/buoy/archive/PR206/realtime/PR206.waves.triaxys.realtime.nc" :
> data := ImportMatrix( url )
```

$$\text{data} := \begin{bmatrix} \text{3335 x 6 Matrix} \\ \text{Data Type: anything} \\ \text{Storage: rectangular} \\ \text{Order: Fortran\_order} \end{bmatrix} \tag{2.1}$$

```
> sigWaveHeight := RemoveNonNumeric( data[ .., 2 ] ) :
> n := numelems( sigWaveHeight )
```

$$n := 3334 \tag{2.2}$$

```
> numBins := 40 :
> p1 := Histogram( sigWaveHeight, size = [ 800, 400 ], bincount = numBins, color = COLOR( RGB, 150/255, 40/255, 27/255 ),
    thickness = 0, style = patchnogrid, transparency = 0.5, background = ColorTools:-Color( "RGB", [ 236/255, 240/255,
    241/255 ] ), axis = [ gridlines = [ 10, color = RGB( 1, 1, 1 ) ] ], axesfont = [ Arial ], labels
    = [ "Significant Wave Height (m)", "Distribution" ], labeldirections = [ horizontal, vertical ], labelfont = [ Arial ], size
    = [ 800, 500 ] ) :
> display( p1 )
```



Maximum Likelihood Estimation

> P := unapply(ProbabilityDensityFunction(Weibull(alpha, beta), t), t, alpha, beta)

$$P := (t, \alpha, \beta) \rightarrow \text{piecewise}\left(t < 0, 0, \frac{\beta\, t^{-1+\beta}\, e^{-\left(\frac{t}{\alpha}\right)^{\beta}}}{\alpha^{\beta}}\right) \quad (3.1$$

> maxLike := $(\alpha, \beta) \rightarrow \text{add}\left(\ln\left(P\left(\text{sigWaveHeight}_i, \alpha, \beta\right)\right), i = 1..n\right)$:

> resultsMLE := Maximize(maxLike(α, β), $\alpha = 0.01..5$, $\beta = 0.01..5$)

$$\text{resultsMLE} := [-252.947777061870255, [\alpha = 1.05355077604419, \beta = 3.93164608045886]] \quad (3.2$$

> p2 := plot(eval(P(t, alpha, beta), resultsMLE[2]) , t = min(sigWaveHeight) ..max(sigWaveHeight), color = black, legend = "Maximum Likelihood Estimation", thickness = 3, legendstyle = [font = [Arial]]) :

Moment Matching

> resultsMM := fsolve([Moment(sigWaveHeight, 1) = Moment(Weibull(α, β), 1), Moment(sigWaveHeight, 2) = Moment(Weibull(α, β), 2)], $\{\alpha = 1, \beta = 1\}$)

(4.1

resultsMM := $\{\alpha = 1.053282268, \beta = 4.217524999\}$ (4.1

```
> p3 := plot( eval( P( t, alpha, beta ), resultsMM ) , t = min( sigWaveHeight ) ..max( sigWaveHeight ), color
    = RGB( 150/255, 40/255, 27/255 ), thickness = 3, legend = "Moment Matching", legendstyle = [font = [Arial]] ) :
```

Results

```
> display( p1, p2, p3, size = [800, 500] )
```